3 - 12 Effect of delta (impulse) on vibrating systems
Find and graph or sketch the solution of the IVP.

3. y'' + 4 y = δ (t - π), y[0] = 8, y'[0] = 0

```
ClearAll["Global`*"]
```

```
e1 = LaplaceTransform[y''[t] + 4 y[t] == DiracDelta[ (t - π)], t, s]
```

$4\,\text{LaplaceTransform}[y[t], t, s] + s^2\,\text{LaplaceTransform}[y[t], t, s] - s\,y[0] - y'[0] == e^{-\pi s}$

```
e2 = e1 /. {y[0] → 8, y'[0] → 0, LaplaceTransform[y[t], t, s] → bigY}
```

$4\,\text{bigY} - 8\,s + \text{bigY}\,s^2 == e^{-\pi s}$

```
e3 = Solve[e2, bigY]
```

$\left\{\left\{\text{bigY} \to \frac{e^{-\pi s}\,(1 + 8\,e^{\pi s}\,s)}{4 + s^2}\right\}\right\}$

```
e4 = e3[[1, 1, 2]]
```

$\frac{e^{-\pi s}\,(1 + 8\,e^{\pi s}\,s)}{4 + s^2}$

```
e5 = InverseLaplaceTransform[e4, s, t]
```

```
8 Cos[2 t] + Cos[t] HeavisideTheta[-π + t] Sin[t]
```

$\text{PossibleZeroQ}\left[\text{Cos}[t]\,\text{Sin}[t] - \frac{1}{2}\,\text{Sin}[2\,t]\right]$

```
True
```

I showed in section 6.3 that **HeavisideTheta** is equivalent to **UnitStep**. Combined with the PZQ above, it makes the green cell equivalent to the text answer.

```
plot1 = Plot[e5, {t, 0, 3 π}, PlotRange → Automatic,
   PlotStyle → {Yellow, Thickness[0.003]}, ImageSize → 250];
plot2 = Plot[8 Cos[2 t] + 1/2 UnitStep[t - π] Sin[2 t], {t, 0, 3 π}, PlotRange →
    Automatic, PlotStyle → {Red, Thickness[0.01]}, ImageSize → 250];
```

```
Show[plot2, plot1]
```



Above: The solution tracks well with that of the text.

5. y'' + y = δ (t - π) - δ (t - 2 π), y[0] = 0, y'[0] = 1

```
ClearAll["Global`*"]
```

```
e1 = LaplaceTransform[
  y''[t] + y[t] == DiracDelta[t - π] - DiracDelta[t - 2 π], t, s]
```

LaplaceTransform[y[t], t, s] + s^2 LaplaceTransform[y[t], t, s] - s y[0] - y′[0] == $-e^{-2\pi s} + e^{-\pi s}$

```
e2 = e1 /. {y[0] → 0, y'[0] → 1, LaplaceTransform[y[t], t, s] → bigY}
```

$-1 + \text{bigY} + \text{bigY}\, s^2 == -e^{-2\pi s} + e^{-\pi s}$

```
e3 = Solve[e2, bigY]
```

$$\left\{\left\{\text{bigY} \to \frac{e^{-2\pi s}\left(-1 + e^{\pi s} + e^{2\pi s}\right)}{1 + s^2}\right\}\right\}$$

```
e4 = e3[[1, 1, 2]]
```

$$\frac{e^{-2\pi s}\left(-1 + e^{\pi s} + e^{2\pi s}\right)}{1 + s^2}$$

```
e5 = InverseLaplaceTransform[e4, s, t]
```

-(-1 + HeavisideTheta[-2 π + t] + HeavisideTheta[-π + t]) Sin[t]

```
e6 = e5 /. {HeavisideTheta[-2 π + t] → 0, HeavisideTheta[-π + t] → 0}
```

Sin[t]

Above: The answer agrees with the text for the subinterval $t < \pi$.

```
e7 = e5 /. {HeavisideTheta[-2 π + t] → 0, HeavisideTheta[-π + t] → 1}
```

0

Above: The answer agrees with the text for the subinterval $\pi < t < 2\pi$.

```
e8 = e5 /. {HeavisideTheta[-2 π + t] → 1, HeavisideTheta[-π + t] → 1}
```

```
-Sin[t]
```

Above: The answer agrees with the text for the subinterval $t > 2\pi$.

```
plot1 = Plot[{e6, e7, e8}, {t, 0, 3 π}, PlotRange → Automatic,
  PlotStyle → {{Blue, Thickness[0.003]}, {RGBColor[0.1, 0.5, 0.3],
     Thickness[0.007]}, {Red, Thickness[0.003]}}, ImageSize → 250,
  Ticks → {{π/2, π, 3 π/2, 2 π, 5 π/2, 3 π}, {-1, -.5, .5, 1}},
  GridLines -> {{π/2, π, 3 π/2, 2 π, 5 π/2, 3 π}, {-1, -.5, .5, 1}}]
```



7. $4 y'' + 24 y' + 37 y = 17 e^{-t} + \delta\left(t - \frac{1}{2}\right)$, $y[0] = 1$, $y'[0] = 1$

```
ClearAll["Global`*"]
```

```
e1 = LaplaceTransform[
  4 y''[t] + 24 y'[t] + 37 y[t] == 17 e^-t + DiracDelta[t - 1/2], t, s]
```

$$37\,\text{LaplaceTransform}[y[t], t, s] + 24\,(s\,\text{LaplaceTransform}[y[t], t, s] - y[0]) + 4\left(s^2\,\text{LaplaceTransform}[y[t], t, s] - s\,y[0] - y'[0]\right) == e^{-s/2} + \frac{17}{1 + s}$$

```
e2 = e1 /. {y[0] → 1, y'[0] → 1, LaplaceTransform[y[t], t, s] → bigY}
```

$$37\,\text{bigY} + 24\,(-1 + \text{bigY}\,s) + 4\left(-1 - s + \text{bigY}\,s^2\right) == e^{-s/2} + \frac{17}{1 + s}$$

```
e3 = Solve[e2, bigY]
```

$$\left\{\left\{\text{bigY} \to \frac{28 + e^{-s/2} + 4 s + \frac{17}{1+s}}{37 + 24 s + 4 s^2}\right\}\right\}$$

```
e4 = e3[[1, 1, 2]]
```

$$\frac{28 + e^{-s/2} + 4 s + \frac{17}{1+s}}{37 + 24 s + 4 s^2}$$

```
e5 = InverseLaplaceTransform[e4, s, t]
```

$$\frac{1}{4} e^{-\frac{i}{4}-\left(3+\frac{i}{2}\right) t} \left(4 e^{\frac{i}{4}} \left(2 i - 2 i e^{i t} + e^{\left(2+\frac{i}{2}\right) t}\right) + i e^{3/2} \left(e^{\frac{i}{2}} - e^{i t}\right) \text{HeavisideTheta}\left[-\frac{1}{2}+t\right]\right)$$

```
e6 = FullSimplify[e5]
```

$$\frac{1}{2} e^{-3 t} \left(2 e^{2 t} - e^{3/2} \text{HeavisideTheta}\left[-\frac{1}{2}+t\right] \text{Sin}\left[\frac{1}{4} (1-2 t)\right] + 8 \text{Sin}\left[\frac{t}{2}\right]\right)$$

```
e7 = Expand[e6]
```

$$e^{-t} - \frac{1}{2} e^{\frac{3}{2}-3 t} \text{HeavisideTheta}\left[-\frac{1}{2}+t\right] \text{Sin}\left[\frac{1}{4} (1-2 t)\right] + 4 e^{-3 t} \text{Sin}\left[\frac{t}{2}\right]$$

$$\text{PossibleZeroQ}\left[\left(e^{-t} - \frac{1}{2} e^{\frac{3}{2}-3 t} \text{Sin}\left[\frac{1}{4} (1-2 t)\right] + 4 e^{-3 t} \text{Sin}\left[\frac{t}{2}\right]\right) - \left(e^{-t} + 4 e^{-3 t} \text{Sin}\left[\frac{1}{2} t\right] + \frac{1}{2} \left(e^{-3 (t-1/2)} \text{Sin}\left[\frac{1}{2} t - \frac{1}{4}\right]\right)\right)\right]$$

```
True
```

Above: By comparison of plots in section 6.3 I decided that **HeavisideTheta** is equivalent to **UnitStep**, the function the text prefers to use. Granted that equivalence, the PZQ above confirms that the green cell is equivalent to the text answer.

```
plot1 = Plot[e7, {t, 0, 1}, PlotRange → Automatic,
   PlotStyle → {Yellow, Thickness[0.002]}, ImageSize → 250];
```

$$\text{plot2} = \text{Plot}\left[e^{-t} + 4 e^{-3 t} \text{Sin}\left[\frac{t}{2}\right] + \frac{1}{2} \text{UnitStep}\left[t - \frac{1}{2}\right] e^{-3 \left(t-\frac{1}{2}\right)} \text{Sin}\left[\frac{t}{2} - \frac{1}{4}\right],\right.$$
$$\{t, 0, 1\}, \text{PlotRange} \to \text{Automatic},$$
$$\left.\text{PlotStyle} \to \{\text{Red}, \text{Thickness}[0.008]\}, \text{ImageSize} \to 250\right];$$

```
Show[plot2, plot1]
```



Note the interesting little gap which seems to exist in the combined plot above.

```
plot3 = Plot[e7, {t, 0.45, 0.55}, PlotRange → Automatic,
   PlotStyle → {Green, Thickness[0.003]}, ImageSize → 250];
```

```
plot5 = Plot[e^-t + 4 e^(-3 t) Sin[t/2] + 1/2 UnitStep[t - 1/2] e^(-3 (t - 1/2)) Sin[t/2 - 1/4],
   {t, 0.45, 0.55}, PlotRange → Automatic,
   PlotStyle → {Red, Thickness[0.01]}, ImageSize → 250];
```

In zoomed view, there is a slight dogleg bend, but no gap. WolframAlpha rules that e7 is continuous on ℝ, so I don't know what the problem is with plotting the superposition.

```
Row[{plot3, plot5, Show[plot5, plot3]}]
```



9. $y'' + 4y' + 5y = (1 - u(t - 10))\, e^{t} - e^{10}\, \delta(t - 10)$,
$y[0] = 0,\ y'[0] = 1$

```
ClearAll["Global`*"]
```

I try again with the method of the last two problems, but this one is harder.

```
e1 = LaplaceTransform[y''[t] + 4 y'[t] + 5 y[t] ==
    (1 - UnitStep[t - 10]) e^t - e^10 DiracDelta[t - 10], t, s]
```

```
5 LaplaceTransform[y[t], t, s] + s^2 LaplaceTransform[y[t], t, s] +
  4 (s LaplaceTransform[y[t], t, s] - y[0]) - s y[0] - y′[0] ==
```

$$-e^{10-10 s} + \frac{1}{-1+s} - \frac{e^{-10(-1+s)}}{-1+s}$$

Above: The Laplace transform is similar to the one in the last problem, as a term containing s has been placed in the denominator of the rhs. I use the same ploy as in the past to isolate the expression I need for the reverse transform.

```
e2 = e1 /. {y[0] → 0, y'[0] → 1, LaplaceTransform[y[t], t, s] → bigY}
```

$$-1 + 5\,\text{bigY} + 4\,\text{bigY}\,s + \text{bigY}\,s^2 == -e^{10-10\,s} + \frac{1}{-1+s} - \frac{e^{-10\,(-1+s)}}{-1+s}$$

```
e3 = Solve[e2, bigY]
```

$$\left\{\left\{\text{bigY} \to \frac{e^{-10\,s}\left(-e^{10} + e^{10\,s}\right)s}{(-1+s)\left(5+4\,s+s^2\right)}\right\}\right\}$$

```
e4 = e3[[1, 1, 2]]
```

$$\frac{e^{-10\,s}\left(-e^{10} + e^{10\,s}\right)s}{(-1+s)\left(5+4\,s+s^2\right)}$$

I try to get a reverse transfrom from the bigY object, in which all subexpressions are real.

```
e5 = InverseLaplaceTransform[e4, s, t]
```

$$\frac{1}{20} e^{(-2-i)\,((2+4\,i)+t)}\left((-1-i)\,e^{10\,i}\left((-3-4\,i) + (4+3\,i)\,e^{2\,i\,t} - (1-i)\,e^{(3+i)\,t}\right) + \left((1-7\,i)\,e^{30+20\,i} + (1+7\,i)\,e^{30+2\,i\,t} - 2\,e^{(3+i)\,((1+3\,i)+t)}\right) \text{HeavisideTheta}[-10+t]\right)$$

But in the result I see there are imaginaries, which, unlike in previous cases, do not disappear after using FullSimplify.

```
e17 = FullSimplify[e5]
```

$$\frac{1}{20} e^{(-2-i)\,((2+4\,i)+t)}\left((-1-i)\,e^{10\,i}\left((-3-4\,i) + (4+3\,i)\,e^{2\,i\,t} - (1-i)\,e^{(3+i)\,t}\right) + \left((1-7\,i)\,e^{30+20\,i} + (1+7\,i)\,e^{30+2\,i\,t} - 2\,e^{(3+i)\,((1+3\,i)+t)}\right) \text{HeavisideTheta}[-10+t]\right)$$

So I take a side step to get rid of the imaginaries. Maybe later I can judge whether this is a wise step.

```
e6 = ComplexExpand[Re[e5]];
```

```
e7 = FullSimplify[e6]
```

$$\frac{1}{10} e^{-2\,t}\left(e^{3\,t} - \text{Cos}[t] + 7\,\text{Sin}[t] + \left(-e^{3\,t} + e^{30}\left(\text{Cos}[10-t] + 7\,\text{Sin}[10-t]\right)\right)\text{UnitStep}[-10+t]\right)$$

Time to bring in the text answer. (In entering the text answer I changed 0.1 to $\frac{1}{10}$ (two occurrences).)

$$e8 = \frac{1}{10}\left(e^{t} + e^{-2t}(-\text{Cos}[t] + 7\,\text{Sin}[t])\right) + \frac{1}{10}\text{UnitStep}[t-10]\left(-e^{-t} + e^{-2t+30}(\text{Cos}[t-10] - 7\,\text{Sin}[t-10])\right)$$

$$\frac{1}{10}\left(e^{t} + e^{-2t}(-\text{Cos}[t] + 7\,\text{Sin}[t])\right) + \frac{1}{10}\left(-e^{-t} + e^{30-2t}(\text{Cos}[10-t] + 7\,\text{Sin}[10-t])\right)\text{UnitStep}[-10+t]$$

I see that the text answer comes up with the correct result for one of the initial conditions. The Mathematica answer also gets past this hurdle.

```
e8t = e8 /. t → 0
```

0

```
e7t = e7 /. t → 0
```

0

```
N[e7t10 = e7 /. t → 11]
```

-1594.81

```
plot1 = Plot[e7, {t, 0, 5}, PlotRange → Automatic,
   PlotStyle → {Yellow, Thickness[0.002]}, ImageSize → 250];
plot2 = Plot[e8, {t, 0, 5}, PlotRange → Automatic,
   PlotStyle → {Red, Thickness[0.014]}, ImageSize → 250];
plot3 = Plot[e17, {t, 0, 5}, PlotRange → Automatic,
   PlotStyle → {Blue, Thickness[0.006]}, ImageSize → 250];
```

Plotting all three of the proposed solutions. On the selected interval they all track one other well.

```
Show[plot2, plot3, plot1]
```



I try subtractive tests but the text answer is not the same as the Mathematica answer. I move on to looking at some more plots.

```
plot3 = Plot[e17, {t, 0, 15},
   PlotRange → {{0, 13}, {-50 000, 50 000}}, PlotStyle →
    {RGBColor[0.4, 0.5, 1], Thickness[0.007]}, ImageSize → 350];
plot4 = Plot[e8, {t, 0, 15}, PlotRange → Automatic,
   PlotStyle → {Red, Thickness[0.014]}, ImageSize → 350];
plot7 = Plot[e7, {t, 0, 15}, PlotRange → Automatic,
   PlotStyle → {White, Thickness[0.003]}, ImageSize → 350];
```

Plotting a slightly longer interval. It seems I have three different functions. The one that has discarded imaginary elements seems to have, for some reason, a slightly smaller range. However, Wolfram Alpha judges it to be continuous on ℝ. In contrast the text function has a jump discontinuity at t=10.

```
Show[plot4, plot3, plot7]
```



Both the Mathematica (real) solution and the text solution meet the second initial condition.

```
dp = D[e8, t];
```

```
dp /. t → 0
```

```
1
```

```
dpm = D[e7, t];
```

```
dpm /. t → 0
```

```
1
```

So if the Mathematica solution meets both initial conditions, is it considered correct?

11. $y'' + 5y' + 6y = u(t - 1) + \delta(t - 2)$, $y[0] = 0$, $y'[0] = 1$

```
ClearAll["Global`*"]
```

```
e1 = LaplaceTransform[
  y''[t] + 5 y'[t] + 6 y[t] == UnitStep[t - 1] + DiracDelta[t - 2], t, s]
```

$$6\,\text{LaplaceTransform}[y[t], t, s] + s^2\,\text{LaplaceTransform}[y[t], t, s] + 5\,(s\,\text{LaplaceTransform}[y[t], t, s] - y[0]) - s\,y[0] - y'[0] == e^{-2s} + \frac{e^{-s}}{s}$$

```
e2 = e1 /. {y[0] → 0, y'[0] → 1, LaplaceTransform[y[t], t, s] → bigY}
```

$$-1 + 6\,\text{bigY} + 5\,\text{bigY}\,s + \text{bigY}\,s^2 == e^{-2s} + \frac{e^{-s}}{s}$$

```
e3 = Solve[e2, bigY]
```

$$\left\{\left\{\text{bigY} \to \frac{e^{-2s}\left(e^{s} + s + e^{2s}\,s\right)}{s\left(6 + 5\,s + s^2\right)}\right\}\right\}$$

```
e4 = e3[[1, 1, 2]]
```

$$\frac{e^{-2s}\left(e^{s} + s + e^{2s}\,s\right)}{s\left(6 + 5\,s + s^2\right)}$$

```
e5 = InverseLaplaceTransform[e4, s, t]
```

$$\frac{1}{6}e^{-3t}\left(6\left(-1 + e^{t}\right) + 6\,e^{4}\left(-e^{2} + e^{t}\right)\text{HeavisideTheta}[-2 + t] + \left(e - e^{t}\right)^{2}\left(2\,e + e^{t}\right)\text{HeavisideTheta}[-1 + t]\right)$$

```
e6 = -e^(-3 t) + e^(-2 t) + 1/6 UnitStep[t - 1] (1 - 3 e^(-2 (t-1)) + 2 e^(-3 (t-1))) +
  UnitStep[t - 2] (e^(-2 (t-2)) - e^(-3 (t-2)))
```

$$-e^{-3t} + e^{-2t} + \left(-e^{-3(-2+t)} + e^{-2(-2+t)}\right)\text{UnitStep}[-2 + t] + \frac{1}{6}\left(1 + 2\,e^{-3(-1+t)} - 3\,e^{-2(-1+t)}\right)\text{UnitStep}[-1 + t]$$

Above: The text answer is entered.

```
plot1 = Plot[e5, {t, 0, 1}, PlotRange → Automatic,
   PlotStyle → {Yellow, Thickness[0.002]}, ImageSize → 250];
plot2 = Plot[e6, {t, 0, 1}, PlotRange → Automatic,
   PlotStyle → {Red, Thickness[0.008]}, ImageSize → 250];
```

```
Show[plot2, plot1]
```



Above: the two plots suggest equality.

```
e7 = e6 /. UnitStep → HeavisideTheta
```

$$-e^{-3t} + e^{-2t} + \left(-e^{-3(-2+t)} + e^{-2(-2+t)}\right) \text{HeavisideTheta}[-2+t] + \frac{1}{6}\left(1 + 2e^{-3(-1+t)} - 3e^{-2(-1+t)}\right) \text{HeavisideTheta}[-1+t]$$

```
FullSimplify[e5 == e7]
True
```

Above: So: If the UnitSteps are exchanged for Heavisides, the answers match.